

EXCEL FOR FINANCIAL ANALYSTS

Make It Match

The financial analyst's playbook for reconciliation in Excel.

EVERY SOLUTION IN EXCEL 2016 **AND** EXCEL 365

BANK RECEIPTS			SALES INVOICES		VERDICT
RCP-5006	39,836.08	— ● —	INV-1006	39,836.08	MATCHED
RCP-5005	12,197.26	— ● —	INV-1005	12,197.26	MATCHED
RCP-5168	28,272.43	— ● —	INV-1012	28,272.45	OUT BY 0.02
RCP-5233	78,964.46	— ● —	3 invoices	78,964.46	GROUPED
RCP-5205	29,654.86	 ●	no reference	—	UNMATCHED
256 RECEIPTS · 293 INVOICES			THE RESIDUAL IS THE DELIVERABLE		

**Eijaz Sheikh**

ASKEIJAZ.COM

Twelve chapters · 25 formula patterns · a 26-tab companion workbook.

Every figure in this book came out of that file – including the five rows above.

Make It Match

The financial analyst's playbook for reconciliation in Excel

One-to-one, one-to-many, many-to-many, and the messy middle.
Every solution in Excel 2016 **and** Excel 365, with a companion
workbook of 26 tabs where every figure in this book was produced.

EIJAZ SHEIKH · askeijaz.com

What this book is for

There's a particular kind of afternoon that every analyst knows.

The ledger says one number. The bank says another. The gap is \$41,000, month-end is tomorrow, and somewhere in eleven hundred rows there's a payment that settled four invoices at once, a customer whose name has a trailing space, and a wire whose reference says nothing more helpful than `TRANSFER`. You have a highlighter and a lot of scrolling ahead of you.

This book is about never having that afternoon again.

It isn't about a reconciliation *tool*. Tools are excellent, and if your company buys one, use it. This is about the far more common situation: you have Excel, you have two lists that ought to agree, and you need a defensible answer by tomorrow. Everything here is built from formulas you can read, explain to an auditor, and hand to whoever takes your job in two years.

Who it's for

You reconcile something every month — sales ledger to bank, subledger to general ledger, payroll to the control account, intercompany to intercompany. You know `SUM` and `VLOOKUP`. You don't consider yourself an Excel person, and you'd rather not have to.

You don't need to be one. You need about nine patterns, and this book is those nine patterns.

The two-formula promise

Every solution in this book is given **twice**.

- **Excel 2016 and earlier.** Because a very large number of finance teams are on locked-down builds, and a book that assumes otherwise is useless to them.
- **Excel 365.** Because when you do have the modern functions, some of these problems stop being problems at all — and you should know what you're missing.

Each pair comes with a plain statement of **which one to use and why**, plus an honest note on what older Excel gives up. Where the modern version is dramatically better, the book says so. Where it's merely shorter, it says that too.

The companion workbook

`AskEijaz-MakeItMatch.xlsx` contains the whole book as a working model.

Everything is built around one invented company — **Harborline Instruments, Inc.** — with three years of sales invoices and bank receipts. The data is deliberately imperfect: amounts stored as text, customer names with trailing spaces, duplicate invoice numbers, references with no invoice number in them, a

payment and its reversal, and cash that arrived on the wrong side of the cutoff. None of that is a mistake. All of it is in the book.

Tabs come in pairs. A **navy** tab works in Excel 2016. A **gold** tab marked 365 does the same job with modern spilled formulas. Open a gold tab in Excel 2016 and you'll see `_xlfn.` names and `#NAME?` errors — that's the version gap made visible, not a broken file.

Cells follow one rule: **blue means type here, everything else is calculated.** Green means a check passed, amber means look at this, red means something is wrong.

How to read it

The chapters build on each other, and the workbook builds with them — by chapter 12 the model reconciles itself and tells you whether it's finished. But each chapter also stands alone, so if you've arrived here because of one specific fire, go straight to it:

If your problem is...	Go to
"The two totals don't agree and I don't know where to start"	Chapter 1
"My lookup returns <code>#N/A</code> for rows I can see are there"	Chapter 2
"I'm matching by hand and it takes all day"	Chapter 3
"Everything is out by pennies"	Chapter 4
"One payment covered five invoices"	Chapters 5 and 6
"I'm hunting for a combination that adds up to \$48,300"	Chapter 7
"The reference says <code>ACME PAYMT</code> and nothing else"	Chapter 8
"Something has been counted twice"	Chapter 9
"The cash arrived on the 1st, not the 31st"	Chapter 10
"My controller wants to know how old the balance is"	Chapter 11
"How do I know when I'm done?"	Chapter 12

Contents

FREE SAMPLE

This file is the contents you are reading now and **one full chapter**, printed exactly as it appears in the book. Entries below without a page number are in the full edition, at askeijaz.com. The companion workbook is not in the sample. **What the other eleven chapters do** is at the end.

Part 1 — Getting the ground straight

1. **What "reconciled" actually means** — matched, explained, unexplained; why the residual is the real deliverable, and the control total that makes the whole thing trustworthy.
2. **Preparing both sides** — cleaning, keys, and why a single trailing space causes more failed matches than every other cause combined.
3. **One-to-one matching, done properly** — replacing the highlighter with a formula that says plainly what it found, what it didn't, and what needs a human.

Part 2 — When one-to-one isn't enough

4. **Tolerance matching** — wire fees, rounding, FX pennies, and setting a threshold you can defend.
5. **One payment, several invoices** — the everyday case that row-by-row matching cannot see.
6. **Many-to-many** — no clean pairing, and how to corner it anyway.
7. **Finding the combination that sums to a number** — the subset-sum hunt, done in bounded Excel, and an honest account of when to stop.

Part 3 — The messy middle

8. **Fuzzy matching on names and references** — pulling 1043 out of ACME PAYMT REF 1043 , and a confidence score that tells you what to review.
9. **Duplicates, reversals, and double-counting** — the quiet killers of an otherwise good reconciliation.
10. **Timing differences** — separating *timing* from *error*, which is what turns an unexplained gap into an explained one.

Part 4 — Closing it out

11. **Aging the residual** — describing what's left in the way a controller actually wants it. 7
12. **The self-proving reconciliation** — control totals, a check register, and making next month take twenty minutes.

Back matter

- **Formula quick reference** — 25 patterns, Excel 2016 and Excel 365 side by side, each one quoted from the companion workbook at the cell address where you'll find it.
- **Workbook tab index** — all 26 tabs, what each holds, and which chapter builds it.
- **The monthly reconciliation checklist** — one page to work down, plus the full list gathered from every chapter.
- **A last word** — what to actually do this month, and where to look when it still doesn't tie.
- **About the author**, and what to read next.
- **Colophon and credits** — how the book was made, rights and licence, and where to send corrections.

A note on the cast

Two people appear throughout. **Megan Whitfield** is a senior financial analyst at Harborline Instruments — she's the one doing the work, and she's meant to be you. **Dale Krieger** is the controller, and he's the one who asks the awkward question about a number you haven't checked yet.

Every figure in this book, and every number quoted in the text, comes from the companion workbook. Nothing is illustrative. If the book says a payment is out by \$58,352.05, you can open the file and find that row.

Chapter 11 • Aging the residual

By the end of this chapter you'll be able to describe what's left over in the way a controller actually wants it — and spot when your aging is measuring your matching rather than your customers.

You'll build	Bucketed aging, an on-account line, and the check that keeps them honest
---------------------	--

You'll need	Chapter 5's grouped matching • chapter 10's cutoff
--------------------	--

Workbook tabs	Ch11 Ageing • Ch11 Ageing 365
----------------------	-------------------------------

11.1 The situation

Megan has \$1,433,764.89 outstanding, and Dale wants to know how worried to be.

She could send him the number. It would be accurate, and it would tell him nothing — because \$1.4 million owed by customers who aren't due to pay yet is a healthy business, and \$1.4 million that's been sitting there for a year is a serious problem, and the number is identical either way.

What Dale needs is the *shape* of it. And the first time Megan builds it, the shape looks alarming: **73.7% over ninety days**.

Then she remembers chapter 5, and checks the on-account line.

11.2 Definitions

Aging. Grouping outstanding balances by how overdue they are. The standard way of describing receivables to anyone who isn't looking at the detail.

Bucket. One age band. *Not yet due, 1–30 days*, and so on. The boundaries are conventions, not laws — pick them to match your credit terms.

Days overdue. Days between the due date and the as-at date. **Not** days since the invoice date, which is a different and much less useful number.

Not yet due. Invoices outstanding but inside their credit terms. Outstanding is not the same as late, and mixing them makes every business look worse than it is.

On account. Cash received but not attached to an invoice. It sits *outside* the buckets and must be reported next to them.

Provision / allowance. The amount you expect not to collect. Usually driven off the aging, which is why the aging has to be right before anyone calculates one.

11.3 What we're building

Five buckets, a total, and one line underneath that changes how you read all of them.

B303	fx	=SUMIF(Age_Bucket,\$A303,Age_Outstanding)		
		A	B	C
302	Ageing summary			
303	Not yet due		148,871.79	10.4%
304	1-30 days		75,580.26	5.3%
305	31-60 days		110,281.85	7.7%
306	61-90 days		42,956.67	3.0%
307	Over 90 days		1,056,074.32	73.7%
308	Total outstanding		1,433,764.89	
309	Cash on account (not yet applied)		568,437.96	

Ch11 Ageing. The same residual, given shape. The cash-on-account line is the one to read first - it is money in the bank that has not been pinned to an invoice, and it makes the old buckets look worse than they are.

11.4 The data

Bucket	Amount	Share
Not yet due	\$148,871.79	10.4%
1–30 days	\$75,580.26	5.3%
31–60 days	\$110,281.85	7.7%
61–90 days	\$42,956.67	3.0%
Over 90 days	\$1,056,074.32	73.7%
Total outstanding	\$1,433,764.89	
<i>Cash on account (not yet applied)</i>	<i>\$568,437.96</i>	

Three-quarters of the balance more than ninety days old would, in a real business, trigger a provision, a difficult board conversation, and probably a change of credit terms.

And most of it isn't real. \$568,437.96 of cash is sitting in the on-account line: money Harborline has *received* and hasn't attached to an invoice. Until it's applied, the invoices it paid keep aging.

WHY THIS MATTERS

An aging built on incomplete matching doesn't just look bad — it makes two errors that point the same way. Old debt is overstated, collections are understated, and the business looks worse than it is on both counts. If your over-90 bucket looks implausible, suspect your matching before you suspect your customers.

11.5 The obvious approach – and why it lets you down

Most homemade agings age from the **invoice date**:

```
=As_At-InvoiceDate
```

This is wrong, and it's wrong in a specific and consistent direction. On 30-day terms, an invoice raised 25 days ago isn't overdue at all — it isn't even due. Aging from the invoice date reports it as 25 days old, and every bucket shifts by roughly your credit period.

The second common error is aging the **invoice** rather than the **balance**. A \$50,000 invoice with \$49,000 paid is a \$1,000 problem, not a \$50,000 one. Age what's outstanding.

The third is the one this chapter opened with: aging before the matching is finished. That doesn't produce a slightly wrong aging — it produces an aging that is largely a picture of your own reconciliation.

11.6 The build

Step 1 · Days overdue, from the due date, on the outstanding balance

```
=IF($G7<=0.005,"",MAX(0,As_At-$C7))
```

Three decisions in one line. Nothing outstanding means no age at all — return "" and the row leaves the aging entirely. `MAX(0, ...)` stops invoices that aren't due yet from showing negative days. And `$C7` is the **due** date.

Step 2 · Put it in a bucket

EXCEL 2016 AND EARLIER

```
=IF($H7="", "",  
    IF($H7=0, "Not yet due",  
        IF($H7<=30, "1-30 days",  
            IF($H7<=60, "31-60 days",  
                IF($H7<=90, "61-90 days", "Over 90 days")))))
```

EXCEL 365

```
=MAP(H8#, LAMBDA(days,  
    IF(days="", "",  
        IF(days=0, "Not yet due",  
            IF(days<=30, "1-30 days",  
                IF(days<=60, "31-60 days",  
                    IF(days<=90, "61-90 days", "Over 90 days"))))))
```

Nested `IF` in ascending order: each test only runs if the previous one failed, so `<=60` already means "and more than 30." Order matters — reverse them and everything lands in the first bucket.

ANALYST TIP

A lookup table beats nested `IF` once you have more than five bands, or when the boundaries change: `=LOOKUP($H7, {0,1,31,61,91}, {"Not yet due", "1-30", "31-60", "61-90", "Over 90"})`. `LOOKUP` needs the breakpoints ascending and gives you boundaries a reader can edit without touching a formula.

Which one, and why. The two are the same logic, and the 365 version wraps it in `MAP` for the reason chapter 3 gave: `H` contains `""` for every settled invoice, and a whole-column test would meet all of those at once. Inside the `LAMBDA`, `days` is one value and the nested `IF` behaves exactly as it does when filled down. *Older Excel gives up nothing here but the fill-down.*

Step 3 · Total each bucket

```
=SUMIF(Age_Bucket, $A303, Age_Outstanding)
```

```
=IF($B$308=0, "", $B303/$B$308)
```

The percentage column is worth the ten seconds. Nobody reads five dollar amounts and forms a view; everybody reads five percentages and forms one immediately.

Step 4 · Report on-account separately, and loudly

```
=SUM(Age_OnAccount)
```

This line is **not** part of the aging total, and putting it inside the buckets would be wrong — it's cash, not debt. But leaving it off the page entirely is worse, because the aging above is unreadable without it.

Amber, directly beneath the total, labelled *Cash on account (not yet applied)*.

REMEMBER

Any aging with a large on-account balance beside it is provisional. You're looking at a picture of the debt *and* a picture of how much matching hasn't been done, and you can't tell which is which until the second number is small.

11.7 Handling the ugly cases

Buckets that don't sum to the total. The commonest aging bug. It happens when a row falls into no bucket — an overpaid invoice, a blank date, a text amount. Control total 2 in chapter 12 exists solely to catch this: `SUM(Age_Outstanding)` must equal the sum of the buckets.

Negative outstanding. An overpaid invoice. It doesn't belong in an age band — you're not owed money that's minus-45-days-late. Route it to the on-account line, which is what the workbook does with `MAX(0, MIN(invoiced, matched))`.

Credit notes. Legitimately negative, and they *do* belong in the aging — a credit note reduces what the customer owes. Age it from its own date.

An invoice with no due date. `MAX(0, As_At - blank)` treats a blank as 1900-01-01 and reports about 46,000 days overdue. Guard the date, or fix it upstream.

Disputes. An invoice under dispute ages exactly like any other and tells you nothing useful. Real credit control functions flag them separately. If you can, add the flag.

The as-at date at a weekend. Chapter 10's point, arriving here: a Saturday cutoff pushes several days of collections into the next period and every bucket looks worse.

11.8 Analyst tips

- **Always show percentages.** Five dollar amounts invite no conclusion; five percentages invite the right one.
- **Match your buckets to your terms.** 30/60/90 is convention. On 45-day terms it's actively misleading, because the first bucket mixes not-yet-due with overdue.

- **Keep last month's aging next to this month's.** The movement between buckets is the actual story. A static aging is a photograph; two are a trend.
- **Report on-account every single time.** It's the number that tells a reader how much to trust the rest of the table.
- **Age the balance, never the invoice.** Partial payments are common and a \$1,000 residue on a \$50,000 invoice is a \$1,000 problem.
- **Check the biggest over-90 items by name.** Two or three usually account for most of the bucket, and they're almost always either a dispute or an unapplied payment.

11.9 Common mistakes and how to fix them

Mistake	What you'll see	The fix
Aging from the invoice date	Everything roughly one credit period too old	Age from the due date
Aging the invoice, not the balance	Part-paid invoices massively overstated	Age Outstanding , not Invoiced
No MAX(0, ...)	Negative days for invoices not yet due	MAX(0, As_At - due)
Bucket tests in the wrong order	Everything in the first bucket	Ascending, each building on the last
Buckets not summing to the total	A silent hole	Control total 2
Overpayments inside the buckets	A negative age band	Route them to on-account
No on-account line	An aging nobody can interpret	Report it directly beneath the total
Aging before matching is finished	A picture of your reconciliation, not your debtors	Chapter 5 first

11.10 Your checklist

- ☐ Age from the due date, on the outstanding balance.
- ☐ Guard against negative days and missing dates.
- ☐ Choose buckets that match your credit terms.
- ☐ Confirm the buckets sum exactly to total outstanding.

- Route overpayments to on-account, not into a bucket.
- Show percentages alongside amounts.
- Report cash on account directly beneath the total.
- If over-90 looks implausible, check on-account before writing anything.
- Put last month's aging beside this month's.
- Name the largest two or three over-90 items.

11.11 In the workbook

Tab	What to look at
Ch11 Ageing	Columns G to J , and the summary block beneath the list
Ch12 Checks	Control total 2, which enforces buckets-equal-total
Ch11 Ageing 365	Ten spilled formulas, and a summary block that generates its own bucket labels
Ch05 One to Many	The chapter that removes most of the over-90 balance

Try this. Look at the over-90 bucket: \$1,056,074.32. Now look at the on-account line: \$568,437.96.

The on-account balance is **54% of the size** of the over-90 bucket. That is not a collections problem sitting on the page — it is cash Harborline has already banked, against invoices that are still quietly aging. A matching problem, and one you now know exactly how to fix.

The rest of the book

You've just read one complete chapter of *Make It Match*, along with the contents and the formula quick reference — the same 25 patterns the full book works through, each given for Excel 2016 and Excel 365.

Chapter 11 was chosen for this sample because it stands on its own. Most of the book doesn't: the chapters build, and the useful part is the machinery underneath the reporting.

What the other eleven chapters do

Part 1 — Getting the ground straight. What "reconciled" actually means, and why the residual is the real deliverable. Then the two chapters that save the most time in practice: cleaning both sides and building keys — a single trailing space causes more failed matches than every other cause combined — and replacing the highlighter with a formula that says plainly what it matched, what it didn't, and what needs a human.

Part 2 — When one-to-one isn't enough. Tolerance matching you can defend to an auditor. One payment covering five invoices, the everyday case row-by-row matching simply cannot see. Many-to-many with no clean pairing. And the subset-sum hunt: finding the combination that adds up to \$48,300, done in bounded Excel, with an honest account of when to stop.

Part 3 — The messy middle. Pulling 1043 out of ACME PAYMT REF 1043 and scoring how confident you are. Duplicates and reversals, the quiet killers of an otherwise good reconciliation. And separating *timing* from *error*, which is what turns an unexplained gap into an explained one.

Part 4 — Closing it out. The chapter you just read is the second-to-last. The last one builds a reconciliation that proves itself — control totals, a check register, and next month taking twenty minutes.

If the aging in chapter 11 looked useful, note what it depended on: chapter 5's grouped matching and chapter 10's cutoff. That is the shape of the whole book.

And the workbook

AskEijaz-MakeItMatch.xlsx isn't in this sample. It's 26 tabs built on Harborline Instruments' three years of invoices and receipts — 293 sales invoices and 256 bank receipts across 36 months — with the mess left in on purpose: amounts stored as text, trailing spaces, duplicate invoice numbers, a payment and its reversal, cash on the wrong side of the cutoff.

Every tab comes twice — a navy one that works in Excel 2016, and a gold 365 twin that does the same job with spilled formulas. Blue means type here. Everything else calculates.

Every figure in the book, and every number quoted in the text, comes out of that file. Nothing is illustrative. Open it and find the row.

Getting it

Make It Match is at askeijaz.com — the PDF and the workbook, downloaded immediately. There are three licences:

Single	One named reader.
Team	Up to ten named people in one organisation.
Department	More than ten, or a whole finance function. Priced on headcount — get in touch and it will be straightforward.

Current prices are on askeijaz.com. This file is meant to be passed around, so it doesn't print a figure that may have moved by the time it reaches you.

Use every formula, pattern and technique in your own work, at your employer or for your clients, without restriction or attribution. Adapt the workbook into your own files freely. That's what it's for.

On refunds, plainly. There isn't a change-of-mind refund, because the files arrive instantly and can't be handed back. That's exactly why this sample exists — read a full chapter, look at the quick reference, and decide before you pay anything. If something is actually wrong with your order, it gets fixed or refunded, no argument.

This sample is yours to pass on

Unlike the book, **this file is meant to be forwarded**. Send it to the colleague who spent last Thursday with a highlighter. Put it on the team channel. That costs nothing, and it's how most people find the book.

Thank you for reading this far.

Eijaz Sheikh · askeijaz.com